

Versal Deformations and Local Hilbert Schemes

NATHAN OWEN ILTEN

ABSTRACT. We provide an overview of the *VersalDeformations* package for *Macaulay2* which computes versal deformations of isolated singularities and local multigraded Hilbert schemes.

INTRODUCTION. Deformation theory provides tools to describe parameter spaces for various algebraic geometric objects including isolated singularities or invertible sheaves on a projective variety; see [Ha]. However, computing such spaces can be quite difficult. The *Macaulay2* [M2] package *VersalDeformations* aims to facilitate such calculations for two concrete deformation problems: versal deformations of isolated singularities, and local (possibly multigraded) Hilbert schemes.

The package *VersalDeformations* provides several functions that may be used to calculate tangent and obstruction spaces for the deformation problems mentioned above. The function `normalMatrix` calculates a basis for any degree of the normal module of some (multi)homogeneous ideal in a polynomial ring. The scripted functor `CT` may be used to calculate bases of the first and second cotangent cohomology modules T_A^1 and T_A^2 of some algebra A over a field k , assuming that these modules are finite dimensional vector spaces. In the homogeneous case, `CT` may also be used to calculate bases of homogeneous pieces of these modules.

The main contribution of the package is the method `versalDeformation`, which uses the Massey product algorithm to iteratively lift solutions of a deformation equation to higher and higher order; we describe this in more detail in the following section. This can be used to find power series descriptions of versal deformations and local Hilbert schemes. Since such a description may not be polynomial, the package provides an interface allowing the user to control at what point the lifting should terminate. The package also implements a more time-consuming lifting algorithm (via the option `SmartLift`) that seeks to minimize the number of higher order terms appearing in the equations for the parameter space.

There are a number of other software packages that provide related functionality. J. Stevens has written scripts for the classic *Macaulay* system to calculate T^1 and T^2 for isolated singularities [S1]. There is a library [Ma] for *Singular* [Sing] by B. Martin that calculates the versal deformation of an isolated singularity or module. B. Hovinen has written a package [Ho] for *Macaulay2* that computes versal deformations of maximal Cohen-Macaulay modules on hypersurfaces. Finally, J. Böhm is developing a package [B] for computations involving deformations of Stanley-Reisner rings.

SOLVING THE DEFORMATION EQUATION. In the following, we briefly describe the Massey product algorithm as we have implemented it. For more details and mathematical background, see [L1], [L2], [S1], or [S2]. For simplicity, we restrict to the case of the versal deformation of an isolated singularity, although our approach for Hilbert schemes is similar.

2010 *Mathematics Subject Classification*. 14D15.
VersalDeformations version 1.0.

First we fix some notation. Let S be a polynomial ring over some field k , and let I be an ideal of S defining a scheme $X = \text{Spec}(S/I)$ with isolated singularity at the origin. Consider a free resolution of S/I , that is an exact sequence $0 \leftarrow S/I \leftarrow S \xleftarrow{F^0} S^m \xleftarrow{R^0} S^\ell \leftarrow \dots$. Let $\varphi_i \in \text{Hom}(S^m/\text{Im}(R^0), S)$ for $1 \leq i \leq n$ represent a basis of $T_{S/I}^1 \cong \text{Hom}(S^m/\text{Im}(R^0), S)/\text{Jac}(F^0)$. We introduce deformation parameters $\mathbf{t} = (t_1, \dots, t_n)$ and consider the map $F^1: S[\mathbf{t}]^m \rightarrow S[\mathbf{t}]$ defined as $F^1 = F^0 + \sum_{i=1}^n t_i \varphi_i$. Let \mathfrak{m} be the ideal generated by t_1, \dots, t_n . It follows that there exists a map $R^1: S[\mathbf{t}]^\ell \rightarrow S[\mathbf{t}]^m$ with $R^1 \equiv R^0 \pmod{\mathfrak{m}}$ satisfying the first order deformation equation $F^1 R^1 \equiv 0 \pmod{\mathfrak{m}^2}$.

Our goal is to lift the above equation to higher order, that is, for each $i > 0$, to find a homomorphism $F^i: S[\mathbf{t}]^m \rightarrow S[\mathbf{t}]$ with $F^i \equiv F^{i-1} \pmod{\mathfrak{m}^i}$ and $R^i: S[\mathbf{t}]^\ell \rightarrow S[\mathbf{t}]^m$ with $R^i \equiv R^{i-1} \pmod{\mathfrak{m}^i}$ satisfying $F^i R^i \equiv 0 \pmod{\mathfrak{m}^{i+1}}$. In general, there are obstructions to doing this, governed by the d -dimensional k vector space $T_{S/I}^2$. Thus, we instead aim to solve

$$(*) \quad (F^i R^i)^t + C^{i-2} G^{i-2} \equiv 0 \pmod{\mathfrak{m}^{i+1}},$$

where $(F^i R^i)^t$ is the transpose of $(F^i R^i)$, and the maps $G^{i-2}: k[\mathbf{t}] \rightarrow k[\mathbf{t}]^d$ and $C^{i-2}: S[\mathbf{t}]^d \rightarrow S[\mathbf{t}]^\ell$ are congruent modulo \mathfrak{m}^i to G^{i-3} and C^{i-3} respectively. Furthermore, we require that G^i and C^i vanish for $i < 0$, and C^0 is of the form $V \cdot D$, where $V \in \text{Hom}(S^d, S^\ell)$ gives representatives of a basis for $T_{S/I}^2$ and $D \in \text{Hom}(S^d, S^d)$ is a diagonal matrix. The matrices G^i now give equations for the miniversal base space of X .

Our implementation solves $(*)$ step by step. Given a solution $(F^i, R^i, G^{i-2}, C^{i-2})$ modulo \mathfrak{m}^{i+1} , the package uses *Macaulay2*'s built in matrix quotients to first solve for F^{i+1} and G^{i-1} (by working over the ring $S[\mathbf{t}]/(I + \text{Im}(G^{i-2})^t + \mathfrak{m}^{i+2})$) and then solve for R^{i+1} and C^{i-1} . For the actual computation, we avoid working over quotient rings involving high powers of \mathfrak{m} by representing the $(F^i, R^i, G^{i-2}, C^{i-2})$ as lists of matrices that keep track of the orders of the parameters t_j involved.

EXAMPLES. We provide two examples: a versal deformation and a multigraded Hilbert scheme. We begin with the classical example of the miniversal deformation of the cone over the rational normal curve of degree 4; see [Pi].

Macaulay2, version 1.4

with packages: ConwayPolynomials, Elimination, IntegralClosure, LLLBases,
PrimaryDecomposition, ReesAlgebra, TangentCone

```
i1 : needsPackage "VersalDeformations";
--loading configuration for package "VersalDeformations" from file ...
i2 : S = QQ[x_0..x_4];
i3 : I = minors(2, matrix{{x_0,x_1,x_2,x_3},{x_1,x_2,x_3,x_4}});
o3 : Ideal of S
i4 : F0 = gens I;
      1      6
o4 : Matrix S  <--- S
i5 : transpose F0
o5 = {-2} | -x_1^2+x_0x_2 |
      {-2} | -x_1x_2+x_0x_3 |
      {-2} | -x_2^2+x_1x_3 |
      {-2} | -x_1x_3+x_0x_4 |
```

```
{-2} | -x_2x_3+x_1x_4 |
{-2} | -x_3^2+x_2x_4 |
      6      1
```

o5 : Matrix S <--- S

We see that the tangent space $T_{S/I}^1$ of the miniversal deformation is four-dimensional, and the obstruction space $T_{S/I}^2$ is three-dimensional.

```
i6 : CT^1(F0)
o6 = {-2} | x_1 x_0 0 0 |
      {-2} | 0 0 0 x_0 |
      {-2} | -x_3 -x_2 0 x_1 |
      {-2} | 0 0 x_2 0 |
      {-2} | -x_4 -x_3 x_3 0 |
      {-2} | 0 0 x_4 -x_3 |
      6      4
```

o6 : Matrix S <--- S

```
i7 : CT^2(F0)
o7 = {-3} | 0 0 0 |
      {-3} | 0 0 0 |
      {-3} | x_3 x_4 0 |
      {-3} | x_2 x_3 0 |
      {-3} | x_1 x_2 0 |
      {-3} | -x_4 0 x_3 |
      {-3} | 0 x_4 x_2 |
      {-3} | 0 x_3 x_1 |
      8      3
```

o7 : Matrix S <--- S

In this example, our algorithm gives a polynomial solution to the deformation equation.

```
i8 : (F,R,G,C) = versalDeformation(F0, Verbose=>2);
Calculating first order deformations and obstruction space
Calculating first order relations
Starting lifting
Order 2
Order 3
Solution is polynomial
i9 : T = ring first G;
i10 : sum G
o10 = | t_2t_3-t_3^2 |
      | t_1t_3      |
      | t_3t_4      |
      3      1
o10 : Matrix T <--- T
i11 : transpose sum F
o11 = {0, -2} | t_1x_1+t_2x_0-x_1^2+x_0x_2 |
      {0, -2} | t_4x_0-x_1x_2+x_0x_3 |
      {0, -2} | -t_1t_4-t_1x_3-t_2x_2+t_4x_1-x_2^2+x_1x_3 |
      {0, -2} | t_2t_3-t_3^2+t_3x_2-x_1x_3+x_0x_4 |
      {0, -2} | t_3t_4-t_1x_4-t_2x_3+t_3x_3-x_2x_3+x_1x_4 |
      {0, -2} | t_3x_4-t_4x_3-x_3^2+x_2x_4 |
```

```
o11 : Matrix T <--- T
      6      1
```

We now consider our second example: the local description of the Hilbert scheme of the diagonal in $\mathbb{P}^2 \times \mathbb{P}^2 \times \mathbb{P}^2$ at the point corresponding to the unique Borel fixed ideal, see [CS] for more details.

```
i12 : S = QQ[x_1..x_3,y_1..y_3,z_1..z_3,Degrees=>{{1,0,0},{1,0,0},
      {1,0,0},{0,1,0},{0,1,0},{0,1,0},{0,0,1},{0,0,1},{0,0,1}}];
```

```
i13 : I=ideal {y_1*z_2, x_1*z_2, y_2*z_1, y_1*z_1, x_2*z_1,
      x_1*z_1, x_1*y_2, x_2*y_1, x_1*y_1, x_2*y_2*z_2};
```

```
o13 : Ideal of S
```

```
i14 : (F,R,G,C) = versalDeformation(gens I, normalMatrix({0,0,0},gens I),
      CT^2({0,0,0},gens I), Verbose=>2);
```

```
Calculating first order relations
Starting lifting
Order 2
Order 3
Order 4
Order 5
Order 6
Solution is polynomial
```

Since we were interested in the multigraded Hilbert scheme, the tangent space is just the degree $(0,0,0)$ component of the normal module of I , and an obstruction space is given by the degree $(0,0,0)$ component of $T_{S/I}^2$. In any case, this multigraded Hilbert scheme is locally cut out by 8 cubics.

```
i15 : T = ring first G;
i16 : sum G
```

```
o16 = | t_2t_3t_4-t_2t_4t_7-t_1t_3t_8+t_1t_7t_8+t_1t_3t_13-t_2t_3t_13-t_1t_7t_13
      | t_1t_3t_4-t_2t_3t_4-t_1t_7t_8+t_2t_7t_8-t_1t_3t_13+t_2t_3t_13+t_1t_7t_13
      | t_1t_3t_16-t_2t_7t_16-t_1t_14t_16+t_2t_14t_16-t_3t_15t_16+t_7t_15t_16
      | t_1t_3t_18-t_2t_7t_18-t_1t_14t_18+t_2t_14t_18-t_3t_15t_18+t_7t_15t_18
      | t_2t_4t_17-t_1t_8t_17+t_1t_13t_17-t_2t_13t_17-t_4t_15t_17+t_8t_15t_17
      | t_2t_4t_18-t_1t_8t_18+t_1t_13t_18-t_2t_13t_18-t_4t_15t_18+t_8t_15t_18
      | t_3t_4t_17-t_7t_8t_17-t_3t_13t_17+t_7t_13t_17-t_4t_14t_17+t_8t_14t_17
      | t_3t_4t_16-t_7t_8t_16-t_3t_13t_16+t_7t_13t_16-t_4t_14t_16+t_8t_14t_16
      -----
      +t_2t_7t_13-t_3t_4t_15+t_4t_7t_15+t_3t_8t_15-t_7t_8t_15 |
      -t_2t_7t_13-t_1t_4t_14+t_2t_4t_14+t_1t_8t_14-t_2t_8t_14 |
```

```
o16 : Matrix T <--- T
      8      1
```

There are in fact 7 irreducible components of the Hilbert scheme that pass through this point.

```
i17 : # primaryDecomposition ideal sum G
o17 = 7
```

REFERENCES.

- [B] J. Böhm, *SRdeformations, A Macaulay2 package for deformations of Stanley-Reisner rings and related computation* (2009), www.math.uiuc.edu/Macaulay2/doc/Macaulay2-1.4/share/Macaulay2/SRdeformations.m2.
- [CS] D. Cartwright and B. Sturmfels, *The Hilbert scheme of the diagonal in a product of projective spaces*, Int. Math. Res. Not. IMRN **9** (2010), 1741 – 1771.
- [Ha] R. Hartshorne, *Deformation theory*, Graduate Texts in Mathematics, vol. 257, Springer, New York, 2010.
- [Ho] B. Hovinen, *Computations of Matrix Factorizations with Macaulay*, JSAG **2** (2010), 6 – 10.
- [L1] O.A. Laudal, *Formal moduli of algebraic structures*, Lecture Notes in Mathematics, vol. 754, Springer, Berlin, 1979.
- [L2] ———, *Matric Massey products and formal moduli. I*, Algebra, algebraic topology and their interactions (Stockholm, 1983), Lecture Notes in Math., vol. 1183, Springer, Berlin, 1986, pp. 218 – 240.
- [M2] D.R. Grayson and M.E. Stillman, *Macaulay2, a software system for research in algebraic geometry*, available at www.math.uiuc.edu/Macaulay2/.
- [Ma] B. Martin, *Computing versal deformations with SINGULAR*, Algorithmic algebra and number theory (Heidelberg, 1997), Springer, Berlin, 1999, pp. 283 – 293.
- [PS] R. Piene and M. Schlessinger, *On the Hilbert scheme compactification of the space of twisted cubics*, Amer. J. Math. **107** (1985), no. 4, 761 – 774.
- [Pi] H.C. Pinkham, *Deformations of algebraic varieties with G_m action*, Astérisque, No. 20, Société Mathématique de France, Paris, 1974.
- [Sing] W. Decker, G.-M. Greuel, G. Pfister, and H. Schönemann, *Singular 3-1-3 — A computer algebra system for polynomial computations* (2011), available at www.singular.uni-kl.de.
- [S1] J. Stevens, *Computing versal deformations*, Experiment. Math. **4** (1995), no. 2, 129 – 144.
- [S2] ———, *Deformations of singularities*, Lecture Notes in Mathematics, vol. 1811, Springer-Verlag, Berlin, 2003.

RECEIVED : 2011-07-20

REVISED : 2012-04-11

ACCEPTED : 2012-06-05

nilten@math.berkeley.edu : Department of Mathematics, University of California, Berkeley, CA 94720, USA.